## EXAMPLE 3: labor-leisure choice

In this problem agent share his endowment of time value on: Leisure and Labour
where labour may be used on production of good X or/and production of good Y

Effective Labour (=Labour Supply*LPROD)is created via 100% of Labour Suply

Market clearing conditions:
for labor is LPROD* LS = CX* X + CY* Y
for leisure is 120 = 1*LS + 1*(120-LS)

Agent has 3 different Demands(good X, good Y, Leisure): maxU(X,Y,L)=ln(X)+ln(Y)+ln(leisure)
s.t. Px*X+Py*Y=PL*LS*LPROD

```
SCALAR          LPROD   AGGREGATE LABOR PRODUCTIVITY /1/,
                CX      COST OF X AT BASE YEAR PRODUCTIVITY /1/,
                CY      COST OF Y AT BASE PRODUCTIVITY /1/;
$ONTEXT
$MODEL:LSUPPLY

$SECTORS:
        X       ! SUPPLY=DEMAND FOR X
        Y       ! SUPPLY=DEMAND FOR Y
        LS      ! LABOR SUPPLY
$COMMODITIES:
        PX      ! MARKET PRICE OF GOOD X
        PY      ! MARKET PRICE OF GOOD Y
        PL      ! MARKET WAGE
        PLS     ! CONSUMER VALUE OF LEISURE
$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT

$PROD:LS
        O:PL    Q:LPROD
        I:PLS   Q:1

$PROD:X
        O:PX    Q:1
        I:PL    Q:CX

$PROD:Y
        O:PY    Q:1
        I:PL    Q:CY

$DEMAND:RA      s:1
        E:PLS   Q:120
        D:PLS   Q:1     P:1
        D:PX    Q:1     P:1
        D:PY    Q:1     P:1

$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY

$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

```
                          LOWER     LEVEL     UPPER    MARGINAL

---- VAR X                  .       40.000     +INF       .
---- VAR Y                  .       40.000     +INF       .
---- VAR LS                 .       80.000     +INF       .
---- VAR PX                 .        1.000     +INF       .
---- VAR PY                 .        1.000     +INF       .
---- VAR PL                 .        1.000     +INF       .
---- VAR PLS                .        1.000     +INF       .
---- VAR RA                 .      120.000     +INF       .
```

**Conclusion:** Consumer ($DEMAND:RA) decides how long to work using "labor production function" ($PROD:LS) and endowment (E:PLS). Leisure*PLS = RA – LS*prod*PL

# Supplement Material to EXAMPLE 3:

We can use this model to evaluate the wage elasticity of labor supply by increasing labor productivity (by 1%)

Elasticity of labor supply is defined as the percentage change in the LS activity. Uncompensated elasticity (ELS) is directly observable, while compensated elasticity is not directly observable.

Uncompensated elasticity is related to Marshallian demand, i.e. utility is maximised given prices and wealth (how demand changes when price changes, holding money income constant). Compensated elasticity is related to Hicksian demand, i.e. expenditure is minimised keeping the utility constant (how demand changes when price changes, holding "real income" or utility constant). The Slutsky relationship: the total (Marshallian) price effect is equal to the sum of the substitution effect (Hicksian price effect) plus an income effect.

```
SCALAR          LPROD   AGGREGATE LABOR PRODUCTIVITY /1/,
                CX      COST OF X AT BASE YEAR PRODUCTIVITY /1/,
                CY      COST OF Y AT BASE PRODUCTIVITY /1/
                LSO     REFERENCE LEVEL OF LABOR SUPPLY/1/
                ELS     uncompensated ELASTICITY OF LABOR supply WRT REAL WAGE/1/;

$ONTEXT
$MODEL:LSUPPLY

$SECTORS:
        X       ! SUPPLY=DEMAND FOR X
        Y       ! SUPPLY=DEMAND FOR Y
        LS      ! LABOR SUPPLY

$COMMODITIES:
        PX      ! MARKET PRICE OF GOOD X
        PY      ! MARKET PRICE OF GOOD Y
        PL      ! MARKET WAGE
        PLS     ! CONSUMER VALUE OF LEISURE

$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT

$PROD:LS
        O:PL    Q:LPROD
        I:PLS   Q:1

$PROD:X
        O:PX    Q:1
        I:PL    Q:CX

$PROD:Y
        O:PY    Q:1
        I:PL    Q:CY

$DEMAND:RA      s:1
        E:PLS   Q:120
        D:PLS   Q:1     P:1
        D:PX    Q:1     P:1
        D:PY    Q:1     P:1

$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
```

- Save last results for LS as LS0

```
LSO = LS.L;
```

- Increase labor productivity

```
LPROD = 1.01;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

ELS = round(100 * (LS.L - LSO) / LSO);
DISPLAY ELS;
```

ROUND means to display integer number. If ELS=0, then without ROUND we will have ELS=0.00000.

```
                              LOWER      LEVEL      UPPER     MARGINAL

---- VAR X                      .        40.400     +INF         .
---- VAR Y                      .        40.400     +INF         .
---- VAR LS                     .        80.000     +INF         .
---- VAR PX                     .         0.990     +INF    -1.772E-7
---- VAR PY                     .         0.990     +INF    -1.772E-7
---- VAR PL                     .         0.990     +INF         .
---- VAR PLS                    .         1.000     +INF    -1.754E-7
---- VAR RA                     .       120.000     +INF    5.2626E-7

-    370 PARAMETER ELS                    =          0.000  uncompensated ELASTIC
                                                            ITY OF LABOR supply W
                                                            RT REAL WAGE
```




```
RA=LS+(RA-LS)=80+(120-80)=120
PL*Lprod*LS=PX*X+PY*Y, i.e.  1 *1  *80 = 1   *40  + 1  *40 becomes
                          0.99*1.01*80 =0.99*40.4 +0.99*40.4
```

**Conclusion:** The increase of labor productivity implies two effects:

1) Income effect: LPROD↑ → PL↑ → income↑, but we keep income=const (by default this is numeraire) → leisure↑
2) Substitution effect: LPROD↑ → output↑ (because we can produce the same amount of X and Y using less time) → PX↓ and PY↓ → demand↑ on X and Y → LS↑

**substitution effect (labor supply↑) = income effect (leisure↑)**

⇓

**LS=const since the above effect exactly balance out**

## Exercise 3A:

One way in which the labor supply elasticity might differ from zero
is if there were income from some other source. Let the consumer be endowed with good x in addition to labor. What x endowment is consistent with a labor supply elasticity (wrt nominal wage) equal to 0.15?

1. First, we have to maxU(X,Y,L)=ln(X)+ln(Y)+ln(L) s.t.
   Px*X+Py*Y=PLS*(EL-L)+Px*EX

2. Using results for (EL-L), find the ETA. If the above calculations are done correctly, you will get the following formula for elasticity (ETA):
   ETA= δLS/δPL * PL0/LS0 =(SHL*PX*EX)/(PL*EL*(1-SHL)-SHL*PX*EX)

3. Calculate EX from the ETA formula

```
SCALAR          LPROD   AGGREGATE LABOR PRODUCTIVITY /1/,
                CX      COST OF X AT BASE YEAR PRODUCTIVITY /1/,
                CY      COST OF Y AT BASE PRODUCTIVITY /1/

                LSO     REFERENCE LEVEL OF LABOR SUPPLY/1/
                ELS     uncompensated ELASTICITY OF LABOR WRT REAL WAGE/1/
                EX      endowment of good x /1/
                EL      endowment of labor and leisure /120/;


$ONTEXT
$MODEL:LSUPPLY

$SECTORS:
        X       ! SUPPLY=DEMAND FOR X
        Y       ! SUPPLY=DEMAND FOR Y
        LS      ! LABOR SUPPLY

$COMMODITIES:
        PX      ! MARKET PRICE OF GOOD X
        PY      ! MARKET PRICE OF GOOD Y
        PL      ! MARKET WAGE
        PLS     ! CONSUMER VALUE OF LEISURE

$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT

$PROD:LS
        O:PL    Q:LPROD
        I:PLS   Q:1

$PROD:X
        O:PX    Q:1
        I:PL    Q:CX

$PROD:Y
        O:PY    Q:1
        I:PL    Q:CY

$DEMAND:RA      s:1
        E:PX    Q:EX
        E:PLS   Q:EL
        D:PLS   Q:1     P:1
        D:PX    Q:1     P:1
        D:PY    Q:1     P:1

$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
```

- First step

```
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

- Second step

```
LSO = LS.L;
LPROD = 1.01;

$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

ELS = round(100 * (LS.L - LSO) / LSO);
DISPLAY ELS;

SCALAR
        ETA       UNCOMPENSATED ELASTICITY OF LABOR SUPPLY wrt NOMINAL wage /0.15/
        SHL       VALUE SHARE OF LEISURE
        LSS        Time endowment /120/
        PRL        real PRICE OF LABOR /1.01/
        PRX        real PRICE OF COMMODITY X /1/;

SHL=(LSS-LS)/LSS
SHL=1/3;
EX=(ETA/(1+ETA)) * LSS *((1-SHL)/SHL) * (PRL/PRX);
DISPLAY EX;
```

- Third step: return to initial productivity (results will be different because EX≠1

```
LPROD = 1;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

- Fourth step: repeat second step

```
LSO = LS.L;
LPROD = 1.01;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
ELS = (LS.L - LSO) / LSO;
OPTION ELS:4;
DISPLAY ELS;
```

Third step solution | Fourth step solution

DUMMY01 Artificial equation for model: LSUPPLY

| | LOWER | LEVEL | UPPER | MARGINAL |
|---|---|---|---|---|
| --- VAR X | . | 18.922 | +INF | . |
| --- VAR Y | . | 50.539 | +INF | . |
| --- VAR LS | . | 69.461 | +INF | . |
| --- VAR PX | . | 0.998 | +INF | -9.253E-8 |
| --- VAR PY | . | 0.998 | +INF | -9.253E-8 |
| --- VAR PL | . | 0.998 | +INF | . |
| --- VAR PLS | . | 0.998 | +INF | -9.253E-8 |
| --- VAR RA | . | 151.317 | +INF | 2.7703E-7 |

| | LOWER | LEVEL | UPPER | MARGINAL |
|---|---|---|---|---|
| - VAR X | . | 19.322 | +INF | . |
| - VAR Y | . | 50.939 | +INF | . |
| - VAR LS | . | 69.565 | +INF | . |
| - VAR PX | . | 0.990 | +INF | -9.385E-8 |
| - VAR PY | . | 0.990 | +INF | -9.385E-8 |
| - VAR PL | . | 0.990 | +INF | . |
| - VAR PLS | . | 1.000 | +INF | -9.292E-8 |
| - VAR RA | . | 151.317 | +INF | 2.7879E-7 |

```
599 PARAMETER ELS                =      0.0015  uncompensated ELASTIC
                                                ITY OF LABOR WRT REAL
                                                WAGE

600 PARAMETER EX                 =      31.617  endowment of good x
```

**Conclusion:** The increase of labor productivity implies ELS=ETA

## Exercise 3B:

We calibrate the labor supply elasticity by changing the utility function from "s:1" to "s:SIGMA" (SIGMA is a scalar representing the elasticity of substitution between x, y, and L in final demand). Find the value of SIGMA consistent with a labor supply elasticity equal to 0.15

```
SCALAR          LPROD   AGGREGATE LABOR PRODUCTIVITY /1/,
                CX      COST OF X AT BASE YEAR PRODUCTIVITY /1/,
                CY      COST OF Y AT BASE PRODUCTIVITY /1/,
                SIGMA    ELASTICITY OF SUBSTITUTION IN CONSUMPTION /1/;

$ONTEXT

$MODEL:LSUPPLY

$SECTORS:
        X       ! SUPPLY=DEMAND FOR X
        Y       ! SUPPLY=DEMAND FOR Y
        LS      ! LABOR SUPPLY

$COMMODITIES:
        PX      ! MARKET PRICE OF GOOD X
        PY      ! MARKET PRICE OF GOOD Y
        PL      ! MARKET WAGE
        PLS     ! CONSUMER VALUE OF LEISURE

$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT

$PROD:LS
        O:PL    Q:LPROD
        I:PLS   Q:1

$PROD:X
        O:PX    Q:1
        I:PL    Q:CX

$PROD:Y
        O:PY    Q:1
        I:PL    Q:CY

$DEMAND:RA      s:SIGMA
        E:PLS   Q:120
        D:PLS   Q:1     P:1
        D:PX    Q:1     P:1
        D:PY    Q:1     P:1

$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
```

- First step

```
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

SCALAR
        LSO     REFERENCE LEVEL OF LABOR SUPPLY
        ELS     uncompensated ELASTICITY OF LABOR WRT REAL WAGE;
```

- Second step

```
LSO = LS.L;
LPROD = 1.01;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

ELS = (LS.L - LSO) / LSO;
DISPLAY ELS;

SCALAR
        ETA     UNCOMPENSATED ELASTICITY OF LABOR SUPPLY wrt nominal wage /0.15/
        SHL     VALUE SHARE OF LEISURE
        LSUP    LABOR SUPPLY /80/
        LEIS    DEMAND FOR LEISURE /40/;
```

```
SHL=1/3;


SIGMA = ETA * (LSUP/LEIS) * (1/(1-SHL)) +1;
DISPLAY SIGMA;
```

- Third step: return to initial productivity (results will be different because EX≠1

```
LPROD = 1;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

- Fourth step: repeat second step

```
LSO = LS.L;
LPROD = 1.01;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
ELS = (LS.L - LSO) / LSO;
DISPLAY ELS;
```

```
                          LOWER     LEVEL     UPPER    MARGINAL

  - VAR X                   .       40.460     +INF       .
  - VAR Y                   .       40.460     +INF       .
  - VAR LS                  .       80.119     +INF       .
  - VAR PX                  .        0.990     +INF   -2.347E-7
  - VAR PY                  .        0.990     +INF   -2.347E-7
  - VAR PL                  .        0.990     +INF       .
  - VAR PLS                 .        1.000     +INF   -2.313E-7
  - VAR RA                  .      120.000     +INF   6.9596E-7

--    580 PARAMETER ELS              =        0.0015  ELASTICITY OF LABOR W
                                                       RT REAL WAGE

    400 PARAMETER SIGMA              =        1.450  ELASTICITY OF SUBSTIT
                                                     UTION IN CONSUMPTION
```

**Conclusion:** CES function allows to get ELS≠0 even with single source of income (labour).

## Supplement Material to EXERCISE 3B:

## How elasticity of substitution influence the results when labor productivity increases by 5%?

**Case 1**: s=1 **Case 2**: s=0

**Case 3:** s<1 **Case 4**: s>1

---

**SOLUTION**

$$\max U(X,Y,R)$$

st $$PX*X+PY*Y+PLS*R=120 \quad \text{or} \qquad (1)$$

$$PX*X+PY*Y=PL*LS*prod \qquad (2)$$

where LS – labor supply

R – leisure

PL – price of labor

PLS – price of leisure

The relationship between PL and PLS can be found from zero-profit condition for LS:

$$PL*Lprod=PLS*1$$

$$\frac{\boldsymbol{PLS}}{\boldsymbol{PL}}=\boldsymbol{Lprod}=1.05 \qquad (3)$$

The relationship between PX and PY can be found from zero-profit condition for X and Y:

$$PX*1=PL*CX, \quad \text{where } CX=1 \rightarrow \frac{PX}{PL}=1$$

$$PY*1=PL*CY, \quad \text{where } CY=1 \rightarrow \frac{PX}{PL}=1$$

$$\Rightarrow \qquad \boldsymbol{PL=PX=PY} \qquad (4)$$

The budget constraint shows the relationship between LS and R:

$$120-PLS*R=PL*LS*Lprod$$

$$\frac{120}{PL}-Lprod*R=LS*Lprod$$

$$\frac{120}{PL}=Lprod\,(R+LS)$$

$$\frac{\boldsymbol{120}}{\boldsymbol{Lprod*PL}}=\boldsymbol{R+LS} \qquad (5)$$

This means that households income is measured by *Lprod*PL* ⇒ by *PLS*. Since default numeraire in MPSGE is households income ⇒ *PLS=1* ⇒ *PL=0.952=PX=PY*

**<u>Case 1:</u>** Cobb-Douglas function

$$\max U = X^{\frac{1}{3}}Y^{\frac{1}{3}}R^{\frac{1}{3}}$$

or linearized version: $$\max lnU = \frac{1}{3}lnX + \frac{1}{3}lnY + \frac{1}{3}lnR$$

$$\frac{MUx}{PX} = \frac{MUy}{PY} = \frac{MUR}{PLS} \quad \Rightarrow \quad \frac{1}{3*X*PX} = \frac{1}{3*Y*PY} = \frac{1}{3*R*PLS}$$

$$\Rightarrow \quad X*PX = Y*PY = R*PLS$$

We can conclude using (4):

$$X = Y$$

The same conclusion we will get for simplified function $U = X*Y*R$. Inserting the above relationship into the budget constraint:

$$3*PX*X = 120$$

Using (3), (4) and PL=1/1.05:

$$PL*X = 40$$

$$X = 42 = Y$$

Using $X*PX = R*PLS$ and PLS=1:

$$PLS*R = 40$$

$$R = 40$$

Finally, using (5):

$$LS = \frac{120}{1.05*0.952} - 40 = 80$$

**part of MPSGE code:**

```
$DEMAND:RA      s:1
        E:PLS   Q:120
        D:PLS   Q:1      P:1
        D:PX    Q:1      P:1
        D:PY    Q:1      P:1
$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

LPROD = 1.05;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

**RESULT:**

```
                         LOWER      LEVEL
---- VAR X                 .        42.000
---- VAR Y                 .        42.000
---- VAR LS                .        80.000
---- VAR PX                .         0.952
---- VAR PY                .         0.952
---- VAR PL                .         0.952
---- VAR PLS               .         1.000
---- VAR RA                .       120.000
```

**Case 2:** Leontief function

$$\max U = min\{X, Y, R\}$$

Substituting (4) in the budget constraint (1):

$$PL * X + PL * X + PLS * X = 120$$

$$X + X + 1.05 * X = \frac{120}{PL}$$

$$X = \frac{120}{3.05 * PL}$$

When PL=0.952:

$$X = 41.311 = Y$$

The last part is to get the labour supply using (5):

$$LS = \frac{120}{1.05 * 0.952} - 41.311 = 78.68$$

**part of MPSGE code:**

```
$DEMAND:RA       s:0
        E:PLS    Q:120
        D:PLS    Q:1      P:1
        D:PX     Q:1      P:1
        D:PY     Q:1      P:1
$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

LPROD = 1.05;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

**RESULT:**

```
                        LOWER
LEVEL

---- VAR X               .        41.311
---- VAR Y               .        41.311
---- VAR LS              .        78.689
---- VAR PX              .         0.952
---- VAR PY              .         0.952
---- VAR PL              .         0.952
---- VAR PLS             .         1.000
---- VAR RA              .       120.000
```

**Case 3 and 4:** CES function

$$\max U = \left(X^{\frac{\sigma-1}{\sigma}} + Y^{\frac{\sigma-1}{\sigma}} + R^{\frac{\sigma-1}{\sigma}}\right)^{\frac{\sigma}{\sigma-1}}$$

$$MUx = \frac{\sigma}{\sigma-1} * U^{\frac{\sigma}{\sigma-1}-1} * \frac{\sigma-1}{\sigma} * X^{\frac{\sigma-1}{\sigma}-1} = U^{\frac{\sigma}{\sigma-1}-1} * X^{\frac{\sigma-1}{\sigma}-1}$$

$$\frac{MUx}{PX} = \frac{MUy}{PY} = \frac{MUR}{PLS} \quad \Rightarrow \quad \frac{U^{\frac{\sigma}{\sigma-1}-1} * X^{\frac{\sigma-1}{\sigma}-1}}{PX} = \frac{U^{\frac{\sigma}{\sigma-1}-1} * Y^{\frac{\sigma-1}{\sigma}-1}}{PY} = \frac{U^{\frac{\sigma}{\sigma-1}-1} * R^{\frac{\sigma-1}{\sigma}-1}}{PLS}$$

Using (4):

$$\frac{MUx}{MUy} = \frac{U^{\frac{\sigma}{\sigma-1}-1} * X^{\frac{\sigma-1}{\sigma}-1}}{U^{\frac{\sigma}{\sigma-1}-1} * Y^{\frac{\sigma-1}{\sigma}-1}} = \frac{PX}{PY} \Rightarrow \frac{X^{\frac{-1}{\sigma}}}{Y^{\frac{-1}{\sigma}}} = \frac{PL}{PL} = 1 \quad \Rightarrow X = Y$$

$$\frac{MUx}{MUR} = \frac{U^{\frac{\sigma}{\sigma-1}-1} * X^{\frac{\sigma-1}{\sigma}-1}}{U^{\frac{\sigma}{\sigma-1}-1} * R^{\frac{\sigma-1}{\sigma}-1}} = \frac{PX}{PLS} \Rightarrow \frac{X^{\frac{-1}{\sigma}}}{R^{\frac{-1}{\sigma}}} = \frac{PL}{PLS} = \frac{1}{1.05} \quad \Rightarrow \text{PL*1.05= PLS}$$

$$R^{-\frac{1}{\sigma}} = 1.05 * X^{-\frac{1}{\sigma}}$$

$$R = 1.05^{-\sigma} * X$$

When $\sigma = 0.5$:

$$R = \frac{X}{1.05^{0.5}}$$

Inserting (4) into the budget constraint (2) when PL=1:

$$2X = (120 - R) * 1.05$$

$$2X + 1.05 * \frac{X}{1.05^{0.5}} = 126$$

$$X = \frac{126}{2 + \frac{1.05}{1.05^{0.5}}} = 41.657 = Y$$

Finally:

$$\text{LS=120-R=120-}\frac{X}{1.05^{0.5}}\text{=}79.347$$

**part of MPSGE code:**

```
$DEMAND:RA        s:0.5
         E:PLS    Q:120
         D:PLS    Q:1      P:1
         D:PX     Q:1      P:1
         D:PY     Q:1      P:1
$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

LPROD = 1.05;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

**RESULT:**

| | LOWER | LEVEL |
|---|---|---|
| ---- VAR X | . | 41.657 |
| ---- VAR Y | . | 41.657 |
| ---- VAR LS | . | 79.347 |
| ---- VAR PX | . | 0.952 |
| ---- VAR PY | . | 0.952 |
| ---- VAR PL | . | 0.952 |
| ---- VAR PLS | . | 1.000 |
| ---- VAR RA | . | 120.000 |

When $\sigma = 2$:

$$R = \frac{X}{1.05^2}$$

$$X = \frac{126}{2 + \frac{1.05}{1.05^2}} = 42.677 = Y$$

$$LS = 120 - R = 120 - \frac{X}{1.05^2} = 81.29$$

**Part of MPSGE code:**

```
$DEMAND:RA       s:2
        E:PLS    Q:120
        D:PLS    Q:1      P:1
        D:PX     Q:1      P:1
        D:PY     Q:1      P:1

$OFFTEXT
$SYSINCLUDE mpsgeset LSUPPLY
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;

LPROD = 1.05;
$INCLUDE LSUPPLY.GEN
SOLVE LSUPPLY USING MCP;
```

**RESULT:**

```
                          LOWER      LEVEL

---- VAR X                  .        42.677
---- VAR Y                  .        42.677
---- VAR LS                 .        81.290
---- VAR PX                 .         0.952
---- VAR PY                 .         0.952
---- VAR PL                 .         0.952
---- VAR PLS                .         1.000
---- VAR RA                 .       120.000
```

**Conclusion**: **(i)** MPSGE defines budget equation as $PX * X + PY * Y + PLS * R = 120$ but not $PX * X + PY * Y = PL * LS * prod$ due to the way how we formulated the model (D: for X,Y,R, but not just for X,Y). **(ii)** Lower substitution possibility between X,Y,R when income is fixed implies decrease of consumption (and labour) while increase of leisure.